```
FILE: STRATIF  FORTRAN  A1  VM/SP HPO REL 4.2 - ELETRONORTE


C          DATA SET STRATIFN   AT LEVEL 050 AS OF 22/04/86
C          DATA SET PEL01230   AT LEVEL 010 AS OF 15/08/85
C
C*************************************************************************
C
C                    THERMAL STRATIFICATION MODEL:
C
C                              STRATIF
C
C                            (VERSION 2)
C
C*************************************************************************
C
C    DEVELOPED BY J.H.G. VERHAGEN EN S. GROOT (VERSION 1),
C    MODIFIED BY J.G.C. SMITS (VERSION 2).
C    DELFT HYDRAULICS LABORATORY,
C    DIVISION OF WATER RESOURCES AND ENVIRONMENT,
C    P.O. BOX 177, 2600MH DELFT, THE NETHERLANDS.
C
C*************************************************************************
C
C    OBJECTIVES:
C
C       1.  COMPUTATION OF MIXING DEPTH (EXCLUDING ENTRAINMENT IF CHANGE
C           OF THE MIXING DEPTH IS INSIGNIFICANT OR NEGATIVE) RESULTING
C           FROM EQUALITY OF THE PRODUCTIONS OF POTENTIAL AND KINETIC
C           ENERGY IN THE EPILIMNION.
C           POTENTIAL ENERGY IS ENLARGED OR REDUCED BY THE NET ENERGY
C           FLUX BETWEEN AIR AND WATER AND BY THE INFLOW OF WATER. IT
C           IS REDUCED BY ENTRAINMENT AND OUTFLOW OF WATER.
C           KINETIC ENERGY IS PRODUCED BY THE WIND AND THE INFLOW.
C       2.  COMPUTATION OF THE AVERAGE TEMPERATURES OF EPILIMNION
C           AND HYPOLIMNION (TWO LAYER MODEL).
C       3.  CALCULATION OF THE DISTRIBUTION OF INFLOW AND OUTFLOW
C           AMONG EPILIMNION AND HYPOLIMNION. THE MODEL ALLOWS THREE
C           INFLOWS FROM DIFFERENT RIVERS, CREEKS OR/AND PRECIPITION
C           AND THREE OUTFLOWS AT DIFFERENT LEVELS.
C
C*************************************************************************
C
C    SPECIFICATIONS:
C
C       1.  THE MAXIMAL NUMBER OF YEARS, WHICH CAN BE SIMULATED IS 20.
C           IN CASE CONTINUATION IS DESIRED, CHANGE THE DIMENSIONS
C           OF THE ARRAYS.
C       2.  THE TIMESTEP IS A MONTH.
C
C*************************************************************************
C
C    INPUT/OUTPUT UNIT NUMBERS:
C
C       1 - INPUT OF MODEL PARAMETERS AND OPTIONS
C       2 - INPUT OF METEOROLOGICAL DATA (TEQ, W, ETC)
C       3 - INPUT OF HYDROLOGICAL DATA AND INFLOW TEMP. (QI, TI, QO)
C       4 - OUTPUT TO BE USED AS INPUT FOR OXY
```

```
FILE: STRATIF  FORTRAN  A1  VM/SP HPO REL 4.2 - ELETRONORTE


C      6 - OUTPUT OF INPUT
C        - OUTPUT OF INTERMEDIATE RESULTS (IF DESIRED)
C        - OUTPUT WITH RESPECT TO THERMAL STRATIFICATION
C      8 - OUTPUT WITH RESPECT TO HYDROLOGY
C      9 - OUTPUT WITH RESPECT TO GEOMETRY AND STRATIFICATION
C
C*****************************************************************
C
C------------------------------------------------- DECLARATIONS ---
C
      DIMENSION A(2),AT(2),H(2),DH(2),VEPI(2),VHYPO(2),VTOT(2)
      DIMENSION CSO(3),HSEL(3),QOS(3)
      DIMENSION QIN  (20,12),QEIN (20,12),QHIN (20,12),
     1          QOUT1(20,12),QOUT2(20,12),QOUT3(20,12),
     1          DEPTH(20,12),HEPI (20,12),HYPO (20,12),
     1          TEPI (20,12),DELT (20,12),TOUT (20,12),
     1          WIND (20,12)
C
C------------------------------------------ DECLARATION OF FORMATS ---
C
  101 FORMAT (65X,I1)
  102 FORMAT (61X,F5.0)
  103 FORMAT (61X,I5)
  201 FORMAT (I5,I3,3F8.0)
  301 FORMAT (I5,I3,7F7.0)
  302 FORMAT (I5,I3,9F7.0)
  401 FORMAT (1H ,'AMAX  = ',F5.0,1X,'KM2',4X,'GAMMA = ',F5.2,8X,
     1        'EXPO  = ',F5.2,8X,'RESL  = ',F5.1,1X,'KM',/)
  402 FORMAT (1H ,'HZERO = ',F5.1,1X,'M',6X,'HMAX  = ',F5.1,1X,'M',6X,
     1        'HSEL1 = ',F5.1,1X,'M',6X,'HSEL2 = ',F5.1,1X,'M',6X,
     1        'HSEL3 = ',F5.1,1X,'M',/)
  403 FORMAT (1H ,'TZERO = ',F5.2,1X,'OC',5X,'DTMP1 = ',F5.2,1X,'OC',5X,
     1        'DTMP2 = ',F5.2,1X,'OC',5X,'DTMP3 = ',F5.2,1X,'OC',/)
  404 FORMAT (1H ,'CE    = ',F5.2,8X,'CVEG  = ',F5.2,8X,'CD    = ',F5.2,
     1        8X,'DISP  = ',F5.2,1X,'M2/D',/)
  405 FORMAT (1H ,'CP    = ',F5.0,1X,'J/KGOC',1X,'RHC   = ',F5.0,1X,
     1        'KG/M3',2X,'RHOA  = ',F5.2,1X,'KG/M3',2X,'BETHA = ',
     1        E9.3,/)
  406 FORMAT (1H ,'BEGIN = ',I5,8X,'END   = ',I5,8X,'DELTT = ',F5.2,
     1        1X,'D',6X,'NTS   = ',I5,8X,'EPSIL = ',F5.2,1X,'M',/)
  407 FORMAT (1H ,'ITITL = ',I5,8X,'INTER = ',I5,8X,'ICUT  = ',I5)
  502 FORMAT (1H1,10X,22H $$$$  $$$$$  $$$$   ,
     1                          24H $$$   $$$$$  $$$  $$$$$,
     1        /1X,10X,22H$         $   $   $   ,
     1                          24H$   $    $     $   $    ,
     1        /1X,10X,22H $$$      $    $$$$   ,
     1                          24H$$$$$    $     $   $$$$ ,
     1        /1X,10X,22H    $     $    $   $   ,
     1                          24H$   $    $     $   $    ,
     1        /1X,10X,22H$$$$      $    $   $   ,
     1                          24H$   $    $     $$$  $    ,//)
  503 FORMAT
     1    (8X,50H**********************************************/
     1     8X,50H     VERSION 1 JUNE 1984, VERSION 2 APRIL 1986    /
     1     8X,50H           DELFT HYDRAULICS LABORATORY            /
```

```
     1     8X,50H    DIVISION OF WATER RESOURCES AND ENVIRONMENT   /
     1     8X,50H**********************************************///)
  504 FORMAT (1H ,3X,'ECW=',E10.3,3X,'EPA=', E10.3,
     1       3X,'EPR=',E10.3,3X,'ECR=',E10.3)
  505 FORMAT (1H ,3X,'NIT=',I3,3X,'CE=',F5.1)
  601 FORMAT (1H1,'DATE     TE     TH  DELT    TO   TEQ   TI1'
     1       ,'   TI2   TI3  WIND   ETC DEPTH  DMIX   RVE   RAT'
     1       ,' ORLOB'//)
  602 FORMAT (1H ,I4,8F6.1,2F6.2,2F6.1,3F6.2)
  603 FORMAT (1H ,1X)
  605 FORMAT (//,1H ,'ERROR CONCERNING DATES IN INPUT')
  801 FORMAT (1H ,'DATE      QIN    QIN1    QIN2    QIN3    QINE',
     1       '   QINE1   QINE2   QINE3    QINH   QINH1   QINH2',
     1       '   QINH3    QOUT   QOUTE   QOUTH COEPI',//)
  802 FORMAT (1H ,I4,15F8.0,F6.2)
  901 FORMAT (1H ,'DATE   DEPTH  HDEPTH    DMIX    AREA   HAREA',
     1       '     RAT    VEPI   VHYPO    VTOT     RVE     RVT',
     1       '   QVERT   QOUT1   QOUT2   QOUT3',//)
  902 FORMAT (1H ,I4,3F8.1,2F8.0,F8.2,3F8.0,2F8.2,4F8.0)
 1000 FORMAT ('QIN ',12F6.0)
 1001 FORMAT ('QEIN',12F6.0)
 1002 FORMAT ('QHIN',12F6.0)
 1003 FORMAT ('DEPT',12F6.1)
 1004 FORMAT ('HEPI',12F6.1)
 1005 FORMAT ('HHYP',12F6.1)
 1006 FORMAT ('QOU1',12F6.0)
 1007 FORMAT ('QOU2',12F6.0)
 1008 FORMAT ('QOU3',12F6.0)
 1009 FORMAT ('WIND',12F6.2)
 1010 FORMAT ('TEPI',12F6.1)
 1011 FORMAT ('TDEL',12F6.1)
 1012 FORMAT ('TOUT',12F6.1)
C
C------------------------------ READ MODEL AND OPTION PARAMETERS ---
C
      READ(1,101) ITITLE
      READ(1,101) INTER
      READ(1,103) IBEGIN
      READ(1,103) IEND
      READ(1,102) DELTAT
      READ(1,103) NTS
      READ(1,102) RESL
      READ(1,102) HZERO
      READ(1,102) TZERO
      READ(1,102) HMAX
      READ(1,102) AMAX
      READ(1,102) EXPO
      READ(1,102) GAMMA
      READ(1,102) CD
      READ(1,102) CVEG
      READ(1,102) DTMP1
      READ(1,102) DTMP2
      READ(1,102) DTMP3
      READ(1,102) DISP
      READ(1,102) EPSIL
```

```
FILE: STRATIF  FORTRAN  A1  VM/SP HPO REL 4.2 - ELETRONORTE


      READ(1,101) IOUT
      IF (IOUT .NE. 2) GOTO 5
      READ(1,102) HSEL(1)
      READ(1,102) HSEL(2)
      READ(1,102) HSEL(3)
    5 CONTINUE
C
C-------------------------------------------------------------------
C
C    HSEL = LEVEL OF OUTLET
C    IOUT = OPTION PARAMETER FOR METHOD DISTRIBUTION OF OUTFLOW
C    IOUT = 1 : ACCORDING TO RELATIVE EPILIMNION VOLUME
C    IOUT = 2 : ACCORDING TO CRITICAL DISCHARGE (FISHER ET AL)
C
C------------------------- ASSIGN VALUES TO PHYSICAL CONSTANTS ---
C
      BETHA =    0.641E-05
      CE    =    1.000
      CP    = 4185.
      G     =    9.810
      RHO   = 1000.
      RHOA  =    1.200
C
C--------------------- WRITE TITLES, INPUT PARAMETERS AND CONSTANTS ---
C
      IF (ITITLE .NE. 1) GOTO 10
      WRITE(6,502)
      WRITE(6,503)
   10 CONTINUE
      WRITE(6,401) AMAX,GAMMA,EXPO,RESL
      WRITE(6,402) HZERO,HMAX,HSEL(1),HSEL(2),HSEL(3)
      WRITE(6,403) TZERO,DTMP1,DTMP2,DTMP3
      WRITE(6,404) CE,CVEG,CD,DISP
      WRITE(6,405) CP,RHO,RHOA,BETHA
      WRITE(6,406) IBEGIN,IEND,DELTAT,NTS,EPSIL
      WRITE(6,407) ITITLE,INTER,IOUT
C
C------------------------------------------------- ADJUST UNITIES ---
C
      DELTAT = DELTAT*86400.
      DELTB  = DELTAT/NTS
      RESL   = RESL*1000.
      AMAX   = AMAX*(10.**6)
      CD     = CD/10000.
C
C------------------------------------ INITIALISATION OF GEOMETRY ---
C
      H(1)      = HZERO
      H(2)      = H(1)
      DH(1)     = H(1)
      A(1)      = AMAX*((1.-GAMMA+(GAMMA*H(1)/HMAX))**EXPO)
      AT(1)     = A(1)
      A(2)      = A(1)
      AT(2)     = AT(1)
      VMAX      = (1.-(1.-GAMMA)**(EXPO+1.))
```

```
      VMAX      = (VMAX*AMAX*HMAX)/(GAMMA*(EXPO+1.))
      VTOT(1)   = (1.-GAMMA+(GAMMA*H(1)/HMAX))**(EXPO+1.)
      VTOT(1)   = VTOT(1)-(1.-GAMMA)**(EXPO+1.)
      VTOT(1)   = (VTOT(1)*AMAX*HMAX)/(GAMMA*(EXPO+1.))
      VTOT(2)   = VTOT(1)
      VEPI(1)   = VTOT(1)
      VEPI(2)   = VEPI(1)
      VHYPO(1)  = 0.0
      VHYPO(2)  = 0.0
      COEPI     = 1.0
C
C------------------------------- INICIAL HOMOGENECUS TEMPERATURE ---
C
      TE = TZERO
      TH = TZERO
      TO = TZERO
C
C-----------------------------------------------------------------
C
C    1. COMPUTATION OF THE MIXING DEPTH ON THE BASIS OF THE EQUALITY OF
C       KINETIC AND POTENTIAL ENERGY PRODUCTIONS
C    2. COMPUTATION OF THE TEMPERATURES OF EPILIMNION AND HYPOLIMNION
C       ON THE BASIS OF THE THERMAL ENERGY BALANCE
C
C-----------------------------------------------------------------
C
      IYEAR = 1
      I     = 0
      IP    = 2
   12 CONTINUE
      IF (I .EQ. 12) IYEAR = IYEAR+1
      IF (I .EQ. 12) IP    = IP   +1
      IF (I .EQ. 12) I     = 0
      I = I+1
      IF (IYEAR .EQ. 21) GOTO 99
C
C------ READ THE METEOROLOGICAL DATA, THE INFLOWS, THE TEMPERATURES ---
C---------------------------------- OF INFLOW(S) AND THE OUTFLOW(S) ---
C
      READ(2,201) LYEAR,LMONTH,TEQ,W,ETC
      IF (LYEAR .LT. 1900 .OR. LMONTH .LE. 0 .OR.
     1    LMONTH .GT. 12) GOTO 99
      READ(3,302) LY,LM,QI1,QI2,QI3,TI1,TI2,TI3,QOS(1),QOS(2),QOS(3)
      QO = QOS(1) + QOS(2) + QOS(3)
      QI = QI1 + QI2 + QI3
      IF (LYEAR .NE. LY .OR. LMONTH .NE. LM) GOTO 95
      LDATE = (LYEAR-1900)*100+LMONTH
C
C---------------------------------------- WRITE TITLES FOR OUTPUT ---
C
      IF (IP .EQ. 2 .AND. I .EQ. 1) WRITE(6,601)
      IF (IP .EQ. 2 .AND. I .EQ. 1) WRITE(8,801)
      IF (IP .EQ. 2 .AND. I .EQ. 1) WRITE(9,901)
      IF (IP .EQ. 2 .AND. I .EQ. 1) IP = 0
C
```

```
FILE: STRATIF  FORTRAN  A1  VM/SP HPO REL 4.2 - ELETRONORTE

C---------- SYSTEMATICAL ADJUSTMENT OF THE TEMPERATURES OF INFLOWS ---
C
C   ALTERNATIVE AS FUNCTION OF DISCHARGE:
C
C      TI = TB + DT*EXP(-QI/QA)
C
C      TB = MINIMAL TEMPERATURE
C      DT = MAXIMAL INCREMENT OF TEMPERATURE
C      QA = THE DISCHARGE FOR WHICH EXP(-1)
C
C------------------------------------------------- ONLY IF DESIRED ---
C
      TI1 = TI1+DTMP1
      TI2 = TI2+DTMP2
      TI3 = TI3+DTMP3
C
C-------------------- VOLUME, DEPTH AND AREA AT END OF NEXT MONTH ---
C
      VTOT(2) = VTOT(1)+DELTAT*(QI1+QI2+QI3-QO)
      H(2)    = ((1.-VTOT(2)/VMAX)*((1.-GAMMA)**(EXPO+1.)))
      H(2)    = H(2)+(VTOT(2)/VMAX)
      H(2)    = (HMAX/GAMMA)*(H(2)**(1./(EXPO+1.))-1.+GAMMA)
      A(2)    = AMAX*((1.-GAMMA+(GAMMA*H(2)/HMAX))**EXPO)
C
C----- DISTRIBUTION INFLOWS AND OUTFLOWS AMONG EPI- AND HYPOLIMNION ---
C
      IF (TE .NE. TH) GOTO 15
      TX = TH
      IF (TI1 .LT. TH .AND. TI1 .GE. 0.1) TX = TI1
      IF (TI2 .LT. TH .AND. TI2 .GE. 0.1) TX = TI2
      IF (TI3 .LT. TH .AND. TI3 .GE. 0.1) TX = TI3
      IF ((TX-TH) .GT. -1.0) GOTO 15
      TH = TX
      VHYPO(2) = DELTAT*QI
   15 CONTINUE
      IF (TE .LE. TH) GOTO 100
      TX1 = TI1
      TX2 = TI2
      TX3 = TI3
      IF (TI1 .GE. TE) TX1 = TE
      IF (TI2 .GE. TE) TX2 = TE
      IF (TI3 .GE. TE) TX3 = TE
      IF (TI1 .LE. TH) TX1 = TH
      IF (TI2 .LE. TH) TX2 = TH
      IF (TI3 .LE. TH) TX3 = TH
      CIEPI1 = (TX1-TH)/(TE-TH)
      CIEPI2 = (TX2-TH)/(TE-TH)
      CIEPI3 = (TX3-TH)/(TE-TH)
      GOTO 110
  100 CONTINUE
      CIEPI1 = 1.0
      CIEPI2 = 1.0
      CIEPI3 = 1.0
  110 CONTINUE
      QIE1 = CIEPI1*QI1
```

```
      QIE2 = CIEPI2*QI2
      QIE3 = CIEPI3*QI3
      QIH1 = QI1-QIE1
      QIH2 = QI2-QIE2
      QIH3 = QI3-QIE3
      QIE  = QIE1+QIE2+QIE3
      QIH  = QI-QIE
C
C------------------------------- DENSITIES FOR POTENTIAL ENERGY ---
C
      RHTR1 = RHO*(1.-(BETHA*((TI1-4.)**2)))
      RHTR2 = RHO*(1.-(BETHA*((TI2-4.)**2)))
      RHTR3 = RHO*(1.-(BETHA*((TI3-4.)**2)))
      RHTE  = RHO*(1.-(BETHA*((TE-4.)**2)))
      RHTH  = RHO*(1.-(BETHA*((TH-4.)**2)))
C
C------------------------------- KINETIC ENERGY PRODUCED BY WIND ---
C
      EKW = ((CD*RHOA)**1.5)*(W**3.75)/(RHO**0.5)
C
C--- POTENTIAL ENERGY FROM NET FLUX OF ENERGY BETWEEN AIR AND WATER ---
C
      ALPHA = 2.*BETHA*(TE-4.)
      EFLUX = ETC*(TEQ-TE)
      EPA   = (G*ALPHA*EFLUX)/CP
C
C---------- BEGINNING OF ITERATION FOR COMPUTATION OF MIXING DEPTH ---
C
      NIT = 0
   13 CONTINUE
      IF (INTER .EQ. 1) WRITE(6,504) EKW,EPA,EPR,EKR
C
C------------------- KINETIC ENERGY PRODUCED BY INFLOW OF RIVER(S) ---
C
      COHYPO = (1-COEPI)
      QTE = (QIE+QO*COEPI)/2
      QTH = (QIH+QO*COHYPO)/2
      F3  = (A(1)+A(2))/AMAX
      F4  = (AT(1)+AT(2))/AMAX
      VHE = QTE*F3*RESL/(VEPI(1)+VEPI(2))
      IF (VHYPO(1)+VHYPO(2) .LE. 0.00001) GOTO 120
      VHH = QTH*F4*RESL/(VHYPO(1)+VHYPO(2))
      GOTO 130
  120 VHH = 0.0
  130 VHORI = ABS(VHE-VHH)
      EKR   = 0.5*RHO*(VHORI**3.)
C
C---------------- POTENTIAL ENERGY PRODUCED BY INFLOW OF RIVER(S) ---
C
      QOE = COEPI*QO
      EPR = ((RHTH-RHTR1)*QIE1 + (RHTH-RHTR2)*QIE2 + (RHTH-RHTR3)*QIE3
     1      - (RHTH-RHTE)*QOE)*4.0*G/(A(1)+A(2)+AT(1)+AT(2))
C
C------ COMPUTE MIXING DEPTH WITHOUT ENTRAINMENT IF CHANGE IS SMALL ---
C------ ------------------- OR NEGATIVE, BUT CHECK VALUE OF DENOMINATOR ---
```

FILE: STRATIF FORTRAN A1 VM/SP HPO REL 4.2 - ELETRONORTE

```
C
      D  = EPA+EPR
      IF (D .LE. 0.) GOTO 17
      DH(2) = (EKW+EKR)/(EPA+EPR)
      D     = DH(2)-DH(1)
      IF (D .LE. EPSIL) GOTO 22
C
C---- COMPUTE MIXING DEPTH WITH ENTRAINMENT, BUT CHECK TEMPERATURES ---
C------------------------------- SOLUTION IS UNDEFINED IF TE = TH ---
C
   17 CONTINUE
      IF (TE .LE. TH) DH(2) = H(2)
      IF (TE .LE. TH) GOTO 22
      DRHO  = RHO*BETHA*((TE-4.)**2-(TH-4.)**2)
      F1    = CVEG*G*DRHO/DELTAT
      F2    = (EPA+EPR-(F1*DH(1)))
      DH(2) = ((SQRT((F2**2)+(4.*F1*(EKW+EKR)))-F2)/(F1*2.))
C
C------------- TESTS WITH RESPECT TO THE PROGRESS OF THE ITERATION ---
C------------------------------------ CHOOSE A NEW HYPOLIMNION DEPTH ---
C
   22 CONTINUE
      IF (DH(2) .GE. H(2)) DH(2) = H(2)
      HHO = HH
      D   = HH-DH(2)
      IF (ABS(D) .LE. 0.1) GOTO 26
      IF (NIT .GT. 19) DH(2) = (HH+HHO)/2.
      IF (NIT .GT. 19) GOTO 26
      HH = (HH + DH(2))/2.
      IF (NIT .GT. 5 .AND. NIT .LT.  7) HH = (HH + HHO)/2.
      IF (NIT .GT. 8 .AND. NIT .LT. 10) HH = (HH + HHO)/2.
C
C----------------- RECOMPUTE VOLUME, DEPTH AND AREA OF HYPOLIMNION ---
C
      HHYPO    = AMAX1(0.0, H(2)-HH)
      VHYPO(2) = (1.-GAMMA+(GAMMA*HHYPO/HMAX))**(EXPO+1.)
      VHYPO(2) = VHYPO(2)-(1.-GAMMA)**(EXPO+1.)
      VHYPO(2) = (VHYPO(2)*AMAX*HMAX)/(GAMMA*(EXPO+1.))
      VEPI(2)  = VTOT(2)-VHYPO(2)
      RVE      = VEPI(2)/VTOT(2)
      AT(2)    = AMAX*((1.-GAMMA+(GAMMA*HHYPO/HMAX))**EXPO)
C
C------------------------------- COMPUTE DITRIBUTION OF OUTFLOW(S) ---
C
      IF (IOUT .NE. 2) GOTO 25
      DO 23 IO = 1,3
      HHYPO = H(2)-DH(2)
      DLEV  = HHYPO-HSEL(IO)
      IF (DLEV .LE. 1.) CSO(IO) = 1.
      IF (DLEV .LE. 1.) GOTO 23
      QCRIT   = ((DLEV)**2.5)*(2.-DLEV/HHYPO)*0.1
      QCRIT   = QOS(IO)/QCRIT
      CSO(IO) = 0.
      IF (QCRIT .LT. 1.) GOTO 23
      CSO(IO) = (2.-DLEV/HHYPO)*0.5
```

```
      QCRIT   = ALOG10(QCRIT)
      QCRTT   = (QCRIT)**2.
      CSO(IO) = CSO(IO)*(QCRTT+10.62*QCRIT)
      CSO(IO) = CSO(IO)/(QCRTT+ 4.88*QCRIT+15.23)
      IF (CSO(IO) .GE. 1.) CSO(IO) = 1.
   23 CONTINUE
      COEPI = 0.
      DO 24 IO = 1,3
      IF (QO .GT. 0.) COEPI = COEPI+QOS(IO)*CSO(IO)/QO
   24 CONTINUE
   25 CONTINUE
      IF (IOUT .NE. 2) COEPI = RVE
C
C-------------------------------- RETURN TO BEGINNING OF ITERATION ---
C
      NIT = NIT+1
      GOTO 13
   26 CONTINUE
      IF (INTER .EQ. 1) WRITE(6,505) NIT,CE
C
C--------- CHECKS ON THE LOWER AND UPPER LIMITS OF THE MIXING DEPTH ---
C
      IF (DH(2) .LE. 5. .AND. H(2) .GE. 5.) DH(2) = 5.
      IF (DH(2) .GE. H(2)) DH(2) = H(2)
C
C------------------------------- RECOMPUTE VOLUME, DEPTH AND AREA ---
C
      HHYPO    = H(2)-DH(2)
      AT(2)    = AMAX*((1.-GAMMA+(GAMMA*HHYPO/HMAX))**EXPO)
      VHYPO(2) = (1.-GAMMA+(GAMMA*HHYPO/HMAX))**(EXPO+1.)
      VHYPO(2) = VHYPO(2)-(1.-GAMMA)**(EXPO+1.)
      VHYPO(2) = (VHYPO(2)*AMAX*HMAX)/(GAMMA*(EXPO+1.))
      VEPI(2)  = VTOT(2)-VHYPO(2)
      RVE      = VEPI(2)/VTOT(2)
      RVT      = VTOT(2)/VMAX
      RAT      = AT(2)/A(2)
C
C---------------------------- RECOMPUTE DISTRIBUTION OF OUTFLOW(S) ---
C
      IF (IOUT .NE. 2) GOTO 29
      DO 27 IO = 1,3
      HHYPO = H(2)-DH(2)
      DLEV  = HHYPO-HSEL(IO)
      IF (DLEV .LE. 1.) CSO(IO) = 1.
      IF (DLEV .LE. 1.) GOTO 27
      QCRIT   = ((DLEV)**2.5)*(2.-DLEV/HHYPO)*0.1
      QCRIT   = QOS(IO)/QCRIT
      CSO(IO) = 0.
      IF (QCRIT .LT. 1.) GOTO 27
      CSO(IO) = (2.-DLEV/HHYPO)*0.5
      QCRIT   = ALOG10(QCRIT)
      QCRTT   = (QCRIT)**2.
      CSO(IO) = CSO(IO)*(QCRTT+10.62*QCRIT)
      CSO(IO) = CSO(IO)/(QCRTT+ 4.88*QCRIT+15.23)
      IF (CSO(IO) .GE. 1.) CSO(IO) = 1.
```

```
   27 CONTINUE
      COEPI = 0.
      DO 28 IO = 1,3
      IF (QO .GT. 0.) COEPI = COEPI+QOS(IO)*CSO(IO)/QO
   28 CONTINUE
   29 CONTINUE
      IF (IOUT .NE. 2) COEPI = RVE
C
C----- COMPUTE OUTFLOWS FROM EPI- AND HYPOLIMNION AND VERTICAL FLOW ---
C------------------------- RESULTING FROM THE CONSERVATION OF MASS ---
C
      COHYPO = 1.-COEPI
      QOE    = COEPI*QO
      QOH    = COHYPO*QO
      QVERT  = QIE-QOE+(VEPI(1)-VEPI(2))/DELTAT
C
C---------------------- IN CASE THE CRITERION OF ORLOB IS EXCEEDED, ---
C---------------------------------- A HOMOGENEOUS SITUATION OCCURS ---
C
      RHTEQ = RHO*(1.-(BETHA*((TEQ-4.)**2)))
      TX = TH
      IF (TI1 .LT. TX) TX = TI1
      RHTM  = RHO*(1.-(BETHA*((TX-4.)**2)))
      DGRAV = (RHTM-RHTEQ)/H(2)
      IF (DGRAV .GT. 0.0) GOTO 60
      ORLOB = 99.
      GOTO 65
   60 RESL1 = RESL*A(2)/AMAX
      HAV   = VTOT(2)/A(2)
      ORLOB = (SQRT(RHO/(G*DGRAV)))*RESL1*(QI+QO)/(2.*HAV*VTOT(2))
      IF (ORLOB .LE. 1.) GOTO 30
   65 VEPI(2)  = VTOT(2)
      VHYPO(2) = 0.
      DH(2) = H(2)
      AT(2) = A(2)
      RVE   = 1.
      RAT   = 0.
      COEPI  = 1.
      COHYPO = 0.
      QOE = QO
      QOH = 0.
      TE = (TE*VEPI(1) + TH*VHYPO(1))/VTOT(1)
      GOTO 33
C
C------ COMPUTATION OF TEMPERATURES FROM THE THERMAL ENERGY BALANCE ---
C
   30 CONTINUE
      IF (VHYPO(1) .LE. 0. .AND. VHYPO(2) .LE. 0.) GOTO 33
      DVTOT  = (VTOT(2)-VTOT(1))/NTS
      DVHYPO = (VHYPO(2)-VHYPO(1))/NTS
      VEPIO  = VEPI(1)
      VHYPOO = VHYPO(1)
      VTOTO  = VTOT(1)
      AO     = A(1)
      IF (QVERT .LT. 0.) GOTO 32
```

```
C
C------------------------------------------- QVERT => 0, DL/DT < 0 ---
C
      DO 75 K=1,NTS
      VTOTN  = VTOTO + DVTOT
      HN     = ((1.-VTOTN/VMAX)*((1.-GAMMA)**(EXPO+1.)))
      HN     = HN+(VTOTN/VMAX)
      HN     = (HMAX/GAMMA)*(HN**(1./(EXPO+1.))-1.+GAMMA)
      AN     = AMAX*((1.-GAMMA+(GAMMA*HN/HMAX))**EXPC)
      VHYPON = VHYPOO + DVHYPO
      VEPIN  = VTOTN - VHYPON
      F5   = (ETC*(AO+AN))/(2.*RHO*CP)
      F8   = QIE1*TI1+QIE2*TI2+QIE3*TI3
      F6   = (VEPIO+VEPIN)/(2.*DELTB)
      TEN  = (F6*TE+F5*(TEQ-TE/2.)+F8-QIE*TE/2.)/
     1       (F6+F5/2.+QIE/2.)
      F8   = QIH1*TI1+QIH2*TI2+QIH3*TI3
      F6   = (VHYPOO+VHYPON)/(2.*DELTB)
      TH   = (F6*TH+F8-QIH*TH/2.+QVERT*(TEN+TE-TH)/2.)/
     1       (F6+QIH/2.+QVERT/2.)
      TE   = TEN
      VTOTO  = VTOTN
      AO     = AN
      VHYPOO = VHYPON
      VEPIO  = VEPIN
   75 CONTINUE
      GOTO 34
C
C------------------------------------------- QVERT < 0, DL/DT > 0 ---
C
   32 CONTINUE
      DO 76 K=1,NTS
      VTOTN  = VTOTO + DVTOT
      HN     = ((1.-VTOTN/VMAX)*((1.-GAMMA)**(EXPO+1.)))
      HN     = HN+(VTOTN/VMAX)
      HN     = (HMAX/GAMMA)*(HN**(1./(EXPO+1.))-1.+GAMMA)
      AN     = AMAX*((1.-GAMMA+(GAMMA*HN/HMAX))**EXPC)
      VHYPON = VHYPOO + DVHYPO
      VEPIN  = VTOTN - VHYPON
      F5   = (ETC*(AO+AN))/(2.*RHO*CP)
      F8   = QIH1*TI1+QIH2*TI2+QIH3*TI3
      F6   = (VHYPOO+VHYPON)/(2.*DELTB)
      THN  = (F6*TH+F8-QIH*TH/2)/
     1       (F6+QIH/2.)
      F8   = QIE1*TI1+QIE2*TI2+QIE3*TI3
      F6   = (VEPIO+VEPIN)/(2.*DELTB)
      TE   = (F6*TE+F5*(TEQ-TE/2.)+F8-QIE*TE/2.+QVERT*(TE-THN-TH)/2.)/
     1       (F6+F5/2.+QIE/2.-QVERT/2.)
      TH = THN
      VTOTO  = VTOTN
      AO     = AN
      VHYPOO = VHYPON
      VEPIO  = VEPIN
   76 CONTINUE
      GOTO 34
```

```
C
C------------------------------------------ HOMOGENEOUS SITUATION ---
C
   33 CONTINUE
      QIE1 = QI1
      QIE2 = QI2
      QIE3 = QI3
      QIE  = QI
      QIH1 = 0.0
      QIH2 = 0.0
      QIH3 = 0.0
      QIH  = 0.0
      QVERT = QIE-QOE+(VEPI(1)-VEPI(2))/DELTAT
      DVTOT = (VTOT(2)-VTOT(1))/NTS
      VEPIO = VEPI(1)
      VTOTO = VTOT(1)
      AO    = A(1)
      DO 77 K=1,NTS
      VTOTN = VTOTO + DVTOT
      HN    = ((1.-VTOTN/VMAX)*((1.-GAMMA)**(EXPO+1.)))
      HN    = HN+(VTOTN/VMAX)
      HN    = (HMAX/GAMMA)*(HN**(1./(EXPO+1.))-1.+GAMMA)
      AN    = AMAX*((1.-GAMMA+(GAMMA*HN/HMAX))**EXPO)
      VEPIN = VTOTN
      F5 = (ETC*(AO+AN))/(2.*RHO*CP)
      F8 = QIE1*TI1+QIE2*TI2+QIE3*TI3
      F6 = (VEPIO+VEPIN)/(2.*DELTB)
      TE = (F6*TE+F5*(TEQ-TE/2.)+F8-QIE*TE/2.)/
     1     (F6+F5/2.+QIE/2.)
      VTOTO = VTOTN
      AO    = AN
      VEPIO = VEPIN
   77 CONTINUE
C
C------------------- SOME FINAL CHECKS AND CALC. OF OUTFLOW TEMPERATURE ---
C
   34 CONTINUE
      IF (VHYPO(2) .LE. 0.) TH = TE
      TO = COEPI*TE+COHYPO*TH
   35 CONTINUE
      IF (TE .EQ. TH) GOTO 37
      IF (TE .GT. TH) GOTO 36
      TE = (VEPI(2)*TE+VHYPO(2)*TH)/VTOT(2)
      TH = TE
      TO = TE
      VHYPO(2) = 0.
      VEPI(2)  = VTOT(2)
      DH(2) = H(2)
      RVE   = 1.
      GOTO 37
   36 CONTINUE
C
C------------------------------ BALANCE INCLUDING VERTICAL DISPERSION ---
C
      IF (DISP .LT. 0.01) GOTO 37
```

```
      F7  = DISP*(TE-TH)/(0.5*H(2))
      TE1 = TE-(F7*(AT(1)+AT(2))*DELTAT)/((VEPI(1)+VEPI(2))*86400.)
      TH1 = TH+(F7*(AT(1)+AT(2))*DELTAT)/((VHYPO(1)+VHYPO(2))*86400.)
      IF (TE1 .GT. TH1) TE = TE1
      IF (TE1 .GT. TH1) TH = TH1
      IF (TE1 .GT. TH1) GOTO 37
      TE = (VEPI(2)*TE+VHYPO(2)*TH)/VTOT(2)
      TH = TE
      TO = TE
      VHYPO(2) = 0.
      VEPI(2)  = VTOT(2)
      DH(2) = H(2)
      COEPI = 1.
      RVE   = 1.
      RAT   = 0.
C
C------------------------------ INITIALISATION OF THE NEXT MONTH ---
C
   37 CONTINUE
      DH(1)    = DH(2)
      H(1)     = H(2)
      A(1)     = A(2)
      AT(1)    = AT(2)
      VTOT(1)  = VTOT(2)
      VEPI(1)  = VEPI(2)
      VHYPO(1) = VHYPO(2)
C
C------------------------------ PREPARATION OF OUTPUT FOR OXY ---
C
      IF (ORLOB .GE. 99.)  ORLOB = 99.
      DEPTH(IYEAR,I) = H(2)
      HYPO (IYEAR,I) = H(2)-DH(2)
      HEPI (IYEAR,I) = DH(2)
      TEPI (IYEAR,I) = TE
      DELT (IYEAR,I) = TE-TH
      TOUT (IYEAR,I) = TO
      QIN  (IYEAR,I) = QI
      QEIN (IYEAR,I) = QIE
      QHIN (IYEAR,I) = QIH
      QOUT1(IYEAR,I) = QOS(1)
      QOUT2(IYEAR,I) = QOS(2)
      QOUT3(IYEAR,I) = QOS(3)
      WIND (IYEAR,I) = W
      HHYPO2 = H(2)-DH(2)
      DLT    = TE-TH
      A2     = A(2)/(10.0**6)
      AT2    = AT(2)/(10.0**6)
      VEPI2  = VEPI(2)/(10.0**6)
      VHYPO2 = VHYPO(2)/(10.0**6)
      VTOT2  = VTOT(2)/(10.0**6)

C
C------------------------------------------------------ WRITE OUTPUT ---
C
      WRITE(6,602) LDATE,TE,TH,DLT,TO,TEQ,TI1,TI2,TI3,W,ETC,H(2),
```

```
     1              DH(2),RVE,RAT,ORLOB
      IF (I .EQ. 12) WRITE(6,603)
      WRITE(8,802) LDATE,QI,QI1,QI2,QI3,QIE,QIE1,QIE2,QIE3,QIH,QIH1,
     1              QIH2,QIH3,QO,QOE,QOH,COEPI
      IF (I .EQ. 12) WRITE(8,603)
      WRITE(9,902) LDATE,H(2),HHYPO2,DH(2),A2,AT2,RAT,VEPI2,VHYPO2,
     1              VTOT2,RVE,RVT,QVERT,QOS(1),QOS(2),QOS(3)
      IF (I .EQ. 12) WRITE(9,603)
C
C---------------------- START A NEW COMPUTATION FOR THE NEXT MONTH ---
C
      GOTO 12
   99 CONTINUE
C
C---------------------------------------- WRITE THE INPUT FOR OXY ---
C
      IYEAR = IYEAR-1
      DO 50 I=1,IYEAR,2
      WRITE(4,1000) (QIN(I,J),J=1,12)
      WRITE(4,1000) (QIN(I+1,J),J=1,12)
      WRITE(4,1001) (QEIN(I,J),J=1,12)
      WRITE(4,1001) (QEIN(I+1,J),J=1,12)
      WRITE(4,1002) (QHIN(I,J),J=1,12)
      WRITE(4,1002) (QHIN(I+1,J),J=1,12)
      WRITE(4,1003) (DEPTH(I,J),J=1,12)
      WRITE(4,1003) (DEPTH(I+1,J),J=1,12)
      WRITE(4,1004) (HEPI(I,J),J=1,12)
      WRITE(4,1004) (HEPI(I+1,J),J=1,12)
      WRITE(4,1005) (HYPO(I,J),J=1,12)
      WRITE(4,1005) (HYPO(I+1,J),J=1,12)
      WRITE(4,1006) (QOUT1(I,J),J=1,12)
      WRITE(4,1006) (QOUT1(I+1,J),J=1,12)
      WRITE(4,1007) (QOUT2(I,J),J=1,12)
      WRITE(4,1007) (QOUT2(I+1,J),J=1,12)
      WRITE(4,1008) (QOUT3(I,J),J=1,12)
      WRITE(4,1008) (QOUT3(I+1,J),J=1,12)
      WRITE(4,1009) (WIND(I,J),J=1,12)
      WRITE(4,1009) (WIND(I+1,J),J=1,12)
      WRITE(4,1010) (TEPI(I,J),J=1,12)
      WRITE(4,1010) (TEPI(I+1,J),J=1,12)
      WRITE(4,1011) (DELT(I,J),J=1,12)
      WRITE(4,1011) (DELT(I+1,J),J=1,12)
      WRITE(4,1012) (TOUT(I,J),J=1,12)
      WRITE(4,1012) (TOUT(I+1,J),J=1,12)
   50 CONTINUE
C
C------------------------------------------------ END OF STRATIF ---
C
      GOTO 96
   95 WRITE(6,605)
   96 CONTINUE
      STOP
      END
```